
However the Scell in CA tests is always configured without SRB/DRB (as they are configured on the Pcell), therefore before any configuration can be done to a future Scell (which in this scenario was earlier a Pcell with RBs configured) the bearers should be released first.

Change 1 
	Function name
	f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell

	Reason for change
	1. Acc. to 36.508 table 4.6.3-9A the PUCCH-COnfigDedicated-v1020 needs to be set to pucch-Format-r10 in case of 2CC or more. In case the target Pcell is same as source Pcell this element was configured during Scell Addition and while being “Need ON” no change would be required. However, if the target Pcell changes, this was never signalled to UE for given cell and therefore it is not acc. to 36.508. So this change is suggesting robust implementation, that is to always signal pucch-Format-r10 whenever HO involves also an Scell in the target. As an alternative, this change could be also limited only to scenarios where Pcell changes.
Correspondingly, the SS needs to be configured the same way as UE in target Pcell.
2. When determining timing for different sub-steps configuration we identified a risk that when a UE starts RACH procedure into the target Pcell earlier than in 40ms (delay between sending of the handover message and v_Timing1 in current TTCN) certain SS configurations(such as: configuration of target cell as a Pcell and C-RNTI based contention resolution on target Pcell) will not be ready and may cause undefined SS behaviour and even test failure.

3. In substep 4a there is configuration of radio bearers at the target cell and it is executed with “TimingInfo_Now” If the HO scenario is such that target Pcell is the source Scell this will lead to a situation where step4a is executed before step4 (which configures source Scell as target Pcell) which leads to undefined behaviour in SS.
4.   In case of the scenario where source Pcell becomes target Scell and source Scell becomes target Pcell (aka HO swap) the Pcell radio bearer configuration still exists on this cell when certain configuration sub-steps are configuring it as the new target Scell. 
However the Scell in CA tests is always configured without SRB/DRB (as they are configured on the Pcell), therefore before any configuration can be done to a future Scell (which in this scenario was earlier a Pcell with RBs configured) the bearers should be released first.

	Summary of change
	5. If during HO there is target Scell, the pucch-Format-r10 is signalled to the UE for the target Pcell and correspondingly it is also configured in the SS
6. We propose to reduce the delay between transmission of the HO signalling message and configuration changes in the SS given in individual sub-steps.
7. Timing in sub-step 4a is changed from NOW to v_Timing2
8. In case of HO swap scenario (Pcell and Scell swap their roles) the radio bearers are released on the old Pcell before new Scell configuration is applied to this cell.

	TTCN module
	Common\EUTRA_CA\EUTRA_CommonFunctions_CA.ttcn

	MCC160 Comment
	


Before change

  function f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell(EUTRA_CellId_Type                     p_SourcePCellId, //Source Pcell

                                                   EUTRA_CellId_Type                     p_SourceSCellId, //Source Scell

                                                   EUTRA_CellId_Type                     p_TargetPCellId, //Target Pcell

                                                   EUTRA_CellId_Type                     p_TargetSCellId, // Target Scell

                                                   template (omit) SCellToAddModList_r10  p_SCellToAddModList := omit,

                                                   template (omit) SCellToReleaseList_r10 p_SCellToReleaseList := omit,

                                                   boolean p_UL_CA := false, // UL CA desired

                                                   template (value) Scell_Capability_Type    p_Scell_Capability := UL_DL,

                                                   template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_DRB1,

                                                   C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,

                                                   template (omit)  MeasConfig           p_MeasConfig    := omit,

                                                   template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                   template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                   template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                   Dsr_TransMax_Type                     p_Dsr_TransMax := n4,

                                                   template (omit) CarrierFreqEUTRA      p_CarrierFreqEutra := omit,

                                                   boolean                               p_AddAdditionalSpectrumEmission := false)  runs on EUTRA_PTC

  {

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;

    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_MeasConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    var SubFrameTiming_Type v_Timing;

    var SubFrameTiming_Type v_Timing1;

    var SubFrameTiming_Type v_Timing2;//@sic R5s150639 sic@

    var SCellIndex_r10  v_SCellIndex;

    var MAC_MainConfig_ScellDeactivationTimer_Type v_DeactTimer := rf32;

    var template (value) Scell_Capability_Type v_Scell_Capability := DlOnly;

    var boolean v_UL_CA := p_UL_CA;

    var boolean v_NoUL_SCell := f_EUTRA_CellInfo_GetNoUL_Frequency(p_TargetSCellId) ; //@sic R5s140083 sic@

    var boolean v_CnfFlag := tsc_NoCnfReq;  /* @sic R5s150922 - Additional changes: v_CnfFlag instead of v_CnfFlag sic@ */

    //@sic R5w150205 sic@

    var template (omit)Pcell_Mode_Type v_Pcell_Mode := omit;

    var EUTRA_FDD_TDD_CellInfo_Type v_SCellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_TargetSCellId);

    var EUTRA_FDD_TDD_CellInfo_Type v_PcellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_TargetPCellId);

    if (v_PcellFDD_TDDInfo.FDD_TDD != v_SCellFDD_TDDInfo.FDD_TDD) { //Scell mode is different than Pcell

      if (v_PcellFDD_TDDInfo.FDD_TDD == FDD) { //Pcell is FDD

        v_Pcell_Mode := cs_Pcell_Mode_FDD;

      } else {                                 //Pcell is TDD

        v_Pcell_Mode := cs_Pcell_Mode_TDD (v_PcellFDD_TDDInfo.TDD_SubframeAssignment);

      }

    }

    if (v_UL_CA) {

      v_Scell_Capability := UL_DL;

    }

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    v_RachProcedureConfig := cs_RachProcedureConfig_CRNTI_HO(f_EUTRA_CellInfo_GetRAR_TA(p_TargetPCellId),

                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetPCellId));

    v_RRCConnectionReconfiguration := f_EUTRA_RRCConnectionReconfiguration_IntraLTE_HO_InterCell(p_SourcePCellId,

                                                                                                 p_TargetPCellId,

                                                                                                 p_C_RNTI,

                                                                                                 p_MeasConfig,

                                                                                                 p_Dsr_TransMax,

                                                                                                 p_CarrierFreqEutra,

                                                                                                 p_AddAdditionalSpectrumEmission,

                                                                                                 cs_RRCConnectionReconfiguration_SCell_AddMod_IEs(p_SCellToAddModList, p_SCellToReleaseList));

    v_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.radioResourceConfigDedicated.physicalConfigDedicated.cqi_ReportConfig := omit;

    v_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.radioResourceConfigDedicated.physicalConfigDedicated.cqi_ReportConfig_r10 := cs_CQI_ReportConfig_r10_DEFAULT;

    v_Timing := f_EUTRA_GetNextSendOccasion(p_SourcePCellId, 700); // 500 ms // Time for schedule of DL RRC message

    //@sic R5s130964 R5s150639 sic@

    v_Timing2 := f_EUTRA_TimingInfoAdd (v_Timing, 20); // time for Target Scell-> Pcell configuration SS configuration

    v_Timing1 := f_EUTRA_TimingInfoAdd (v_Timing, 40); // time for Target Pcell SS configuration
    // The 40 ms is sufficient for 4 retransmissions and less than 15 [RRC processing delay] + 100 [5 retransmissions of PRACH preamble on trarget pCell]

    // Prepare target Pcell

    // Step 1: stop periodic TA on source Pcell

    f_EUTRA_SS_CommonCellConfig(p_SourcePCellId, cas_ULGrantAllocation_REQ(p_SourcePCellId,cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 2: Configure no RACH response in target Pcell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetPCellId, p_C_RNTI, cs_RachProcedureConfig_NoResponse(f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetPCellId)), cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    // Step 3: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourcePCellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),v_RRCConnectionReconfiguration));

    // Step 4: optional if target Pcell is Source Scell

    if (p_SourceSCellId == p_TargetPCellId) {  //Else, the target Pcell is expected to be configured with SRB1-3 and DRB already

      f_EUTRA_CellConfig_ScellToPcell_Def(p_TargetPCellId, cs_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number)); //@sic R5s150356 R5s150639 so as to have valid PDCP SQN for Ciphering Activation in step 8 sic@

    }

    // Step 4a: Configuration of DRBs at the target cell

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetPCellId, p_DrbConfigList);       /* @sic R5s150338 - dynamic DRB configuration sic@ */
    // Step 5: Transfer of the PDCP Count for AM DRBs from source to target cell

    f_EUTRA_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_DRB_IdentityList.AMonly, cs_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number)); //@sic R5s150356 R5s150639 so as to have valid PDCP SQN for Ciphering Activation in step 8 sic@

    // Step 6: Tell the SS about the HO

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_PdcpHandoverInit_REQ(p_SourcePCellId, p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number)));

    // Step 7: Configure C-RNTI based contention resolution in target Pcell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetPCellId, p_C_RNTI, v_RachProcedureConfig, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetPCellId, p_C_RNTI);

    // Step 8: activate RRC security at SS target Pcell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Common(p_TargetPCellId,

                                                              v_Auth_Params,

                                                              v_NextHopChainingCount,

                                                              cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number),

                                                              v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 9: configure DRX and MeasGap at the target P and S cells (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourcePCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourcePCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetPCellId, v_DrxCtrl, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

      f_EUTRA_SS_DrxCtrlConfig(p_TargetSCellId, v_DrxCtrl, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetPCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)),cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetSCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)),cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    }

    // Step 10: configure UL grant configuration on target Pcell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_ULGrantAllocation_REQ(p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    // Step 11: Configure target PCell as PCell

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_PCellConfig_REQ(p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), {p_TargetSCellId}, v_CnfFlag));

    // Step 12: Configure target Scell

    if(isvalue (p_SCellToAddModList)) {

      v_SCellIndex := valueof(p_SCellToAddModList[0].sCellIndex_r10);

    } else {

      v_SCellIndex := 1;

    }

    //if Scell change does not happen, steps 1-2 are still needed due to possible change in C-RNTI

    // Save new C-RNTI in cell configuration for p_SCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetSCellId, p_C_RNTI);

    if ( not v_NoUL_SCell) //@sic R5s140083 sic@

    {

    // Step 12.1: Configure C-RNTI based contention resolution in S-cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetSCellId, p_C_RNTI, v_RachProcedureConfig, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), v_CnfFlag);

    // Step 12.2: configure def UL grant configuration and no auto TA transmision on target Scell

    f_EUTRA_SS_CommonCellConfig(p_TargetSCellId, cas_ULGrantAllocation_REQ(p_TargetSCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), cs_DciInfo_CcchDcchDtchUL_Def, cs_PUCCH_Synch_None, cs_UL_GrantConfig_Def, v_CnfFlag));

    };

    // Step 12.3: Configure SCell as Scell with new Pcell association

    f_EUTRA_SS_CommonCellConfig(p_TargetSCellId, cas_SCellConfig_REQ(p_TargetSCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), cs_SCellConfig(p_TargetPCellId, v_SCellIndex, p_Scell_Capability, v_DeactTimer,-,-,v_Pcell_Mode), v_CnfFlag));//@sic R5w150205 sic@

    // Step 13:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetPCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Step 14: configure periodic TA on target cell

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_PUCCH_Synch_Config_REQ(p_TargetPCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 15: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_PdcpHandoverComplete_REQ(p_TargetPCellId));

    // Step 16: configure RACH procedure for target cell back to default

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetPCellId, -, -, -, -, -, v_CnfFlag);

    if (p_SourcePCellId != p_TargetSCellId) {  //Source cell is no more a Carrier component

      // Step 17: Re-establish SRBs and release DRB(s) in source cell

      f_EUTRA_SS_SRBs_DRBs_Reset(p_SourcePCellId, -, -, -, v_CnfFlag); // @sic R5s150338: no reconfiguration of DRBs; R5s150922 - Additional changes: f_EUTRA_SS_SRBs_DRBs_Reset sic@

      // Step 18: Release DRX and MeasGap configuration

      f_EUTRA_SS_DrxCtrlConfig(p_SourcePCellId, cs_DrxCtrl_None, -, v_CnfFlag);

      f_EUTRA_SS_MeasGapCtrlConfig(p_SourcePCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);

      //Step 19: Configure source Pcell as normal cell

      f_EUTRA_SS_CommonCellConfig(p_SourcePCellId, cas_ServingCellRelease_REQ(p_SourcePCellId, cs_TimingInfo_Now, v_CnfFlag));

    }

    if (p_SourceSCellId != p_TargetSCellId and p_SourceSCellId != p_TargetPCellId) {

      // No RACH configuration as Already configured for C-RNTI based contention along with Pcell

      //No update of C-RNTI

      //Step 20

      //  Release DRX and MeasGap at the S cell (if necessary)

      if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceSCellId)) {

        f_EUTRA_SS_DrxCtrlConfig(p_SourceSCellId, cs_DrxCtrl_None, -, v_CnfFlag);

      }

      f_EUTRA_SS_MeasGapCtrlConfig(p_SourceSCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);

      //Step 21

      //  Configure SCell as normal cell

      f_EUTRA_SS_CommonCellConfig(p_SourceSCellId, cas_ServingCellRelease_REQ(p_SourceSCellId, cs_TimingInfo_Now, v_CnfFlag));

      //  configure Default grant configuration on S cell

      f_EUTRA_SS_CommonCellConfig(p_SourceSCellId, cas_ULGrantAllocation_REQ(p_SourceSCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, -, v_CnfFlag));

    }

  }
After change

  function f_EUTRA_508RRC_IntraLTE_CA_HO_InterCell(EUTRA_CellId_Type                     p_SourcePCellId, //Source Pcell

                                                   EUTRA_CellId_Type                     p_SourceSCellId, //Source Scell

                                                   EUTRA_CellId_Type                     p_TargetPCellId, //Target Pcell

                                                   EUTRA_CellId_Type                     p_TargetSCellId, // Target Scell

                                                   template (omit) SCellToAddModList_r10  p_SCellToAddModList := omit,

                                                   template (omit) SCellToReleaseList_r10 p_SCellToReleaseList := omit,

                                                   boolean p_UL_CA := false, // UL CA desired

                                                   template (value) Scell_Capability_Type    p_Scell_Capability := UL_DL,

                                                   template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_DRB1,

                                                   C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,

                                                   template (omit)  MeasConfig           p_MeasConfig    := omit,

                                                   template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                   template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                   template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                   Dsr_TransMax_Type                     p_Dsr_TransMax := n4,

                                                   template (omit) CarrierFreqEUTRA      p_CarrierFreqEutra := omit,

                                                   boolean                               p_AddAdditionalSpectrumEmission := false)  runs on EUTRA_PTC

  {

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var template (value) RachProcedureConfig_Type v_RachProcedureConfig;

    var template (value) DL_DCCH_Message v_RRCConnectionReconfiguration;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_MeasConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    var SubFrameTiming_Type v_Timing;

    var SubFrameTiming_Type v_Timing1;

    var SubFrameTiming_Type v_Timing2;//@sic R5s150639 sic@

    var SCellIndex_r10  v_SCellIndex;

    var MAC_MainConfig_ScellDeactivationTimer_Type v_DeactTimer := rf32;

    var template (value) Scell_Capability_Type v_Scell_Capability := DlOnly;

    var boolean v_UL_CA := p_UL_CA;

    var boolean v_NoUL_SCell := f_EUTRA_CellInfo_GetNoUL_Frequency(p_TargetSCellId) ; //@sic R5s140083 sic@

    var boolean v_CnfFlag := tsc_NoCnfReq;  /* @sic R5s150922 - Additional changes: v_CnfFlag instead of v_CnfFlag sic@ */

    var template (value) PUCCH_ConfigDedicated_v1020 v_PUCCH_ConfigDedicated; //CH1
    //@sic R5w150205 sic@

    var template (omit)Pcell_Mode_Type v_Pcell_Mode := omit;

    var EUTRA_FDD_TDD_CellInfo_Type v_SCellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_TargetSCellId);

    var EUTRA_FDD_TDD_CellInfo_Type v_PcellFDD_TDDInfo := f_EUTRA_CellInfo_GetFDD_TDD_Info(p_TargetPCellId);

    if (v_PcellFDD_TDDInfo.FDD_TDD != v_SCellFDD_TDDInfo.FDD_TDD) { //Scell mode is different than Pcell

      if (v_PcellFDD_TDDInfo.FDD_TDD == FDD) { //Pcell is FDD

        v_Pcell_Mode := cs_Pcell_Mode_FDD;

      } else {                                 //Pcell is TDD

        v_Pcell_Mode := cs_Pcell_Mode_TDD (v_PcellFDD_TDDInfo.TDD_SubframeAssignment);

      }

    }

    if (v_UL_CA) {

      v_Scell_Capability := UL_DL;

    }

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    v_RachProcedureConfig := cs_RachProcedureConfig_CRNTI_HO(f_EUTRA_CellInfo_GetRAR_TA(p_TargetPCellId),

                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetPCellId));

    v_RRCConnectionReconfiguration := f_EUTRA_RRCConnectionReconfiguration_IntraLTE_HO_InterCell(p_SourcePCellId,

                                                                                                 p_TargetPCellId,

                                                                                                 p_C_RNTI,

                                                                                                 p_MeasConfig,

                                                                                                 p_Dsr_TransMax,

                                                                                                 p_CarrierFreqEutra,

                                                                                                 p_AddAdditionalSpectrumEmission,

                                                                                                 cs_RRCConnectionReconfiguration_SCell_AddMod_IEs(p_SCellToAddModList, p_SCellToReleaseList));

    v_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.radioResourceConfigDedicated.physicalConfigDedicated.cqi_ReportConfig := omit;

    v_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.radioResourceConfigDedicated.physicalConfigDedicated.cqi_ReportConfig_r10 := cs_CQI_ReportConfig_r10_DEFAULT;

    //CH1:

    if (v_PcellFDD_TDDInfo.FDD_TDD == TDD) {

        v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_2TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX, cs_N1PUCCH_AN_CS_r10_FDD_2TX);

      }

      else { //FDD

        v_PUCCH_ConfigDedicated  := cs_508_PUCCH_ConfigDedicated_v1020_1TX(cs_N1PUCCH_AN_CS_r10_FDD_1TX);

      }

    if (ispresent(p_TargetSCellId)) {

    v_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.radioResourceConfigDedicated.physicalConfigDedicated.pucch_ConfigDedicated_v1020 := v_PUCCH_ConfigDedicated;

    }
    v_Timing := f_EUTRA_GetNextSendOccasion(p_SourcePCellId, 700); // 500 ms // Time for schedule of DL RRC message

    //@sic R5s130964 R5s150639 sic@

    //CH2: 

    v_Timing2 := f_EUTRA_TimingInfoAdd (v_Timing, 10); // time for Target Scell-> Pcell configuration SS configuration

    v_Timing1 := f_EUTRA_TimingInfoAdd (v_Timing, 20); // time for Target Pcell SS configuration
    // The 40 ms is sufficient for 4 retransmissions and less than 15 [RRC processing delay] + 100 [5 retransmissions of PRACH preamble on trarget pCell]

    // Prepare target Pcell

    // Step 1: stop periodic TA on source Pcell

    f_EUTRA_SS_CommonCellConfig(p_SourcePCellId, cas_ULGrantAllocation_REQ(p_SourcePCellId,cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 2: Configure no RACH response in target Pcell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetPCellId, p_C_RNTI, cs_RachProcedureConfig_NoResponse(f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetPCellId)), cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    // Step 3: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourcePCellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number),v_RRCConnectionReconfiguration));

    // Step 4: optional if target Pcell is Source Scell

    if (p_SourceSCellId == p_TargetPCellId) {  //Else, the target Pcell is expected to be configured with SRB1-3 and DRB already

      f_EUTRA_CellConfig_ScellToPcell_Def(p_TargetPCellId, cs_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number)); //@sic R5s150356 R5s150639 so as to have valid PDCP SQN for Ciphering Activation in step 8 sic@

    }

    // Step 4a: Configuration of DRBs at the target cell

    //CH3:

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetPCellId, p_DrbConfigList, cs_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number));       /* @sic R5s150338 - dynamic DRB configuration sic@ */

    // Step 5: Transfer of the PDCP Count for AM DRBs from source to target cell

    f_EUTRA_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_DRB_IdentityList.AMonly, cs_TimingInfo(v_Timing2.SFN.Number, v_Timing2.Subframe.Number)); //@sic R5s150356 R5s150639 so as to have valid PDCP SQN for Ciphering Activation in step 8 sic@

    // Step 6: Tell the SS about the HO

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_PdcpHandoverInit_REQ(p_SourcePCellId, p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number)));

    // Step 7: Configure C-RNTI based contention resolution in target Pcell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetPCellId, p_C_RNTI, v_RachProcedureConfig, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetPCellId, p_C_RNTI);

    // Step 8: activate RRC security at SS target Pcell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Common(p_TargetPCellId,

                                                              v_Auth_Params,

                                                              v_NextHopChainingCount,

                                                              cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number),

                                                              v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 9: configure DRX and MeasGap at the target P and S cells (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourcePCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourcePCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetPCellId, v_DrxCtrl, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

      f_EUTRA_SS_DrxCtrlConfig(p_TargetSCellId, v_DrxCtrl, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetPCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)),cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetSCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)),cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number));

    }

    // Step 10: configure UL grant configuration on target Pcell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_ULGrantAllocation_REQ(p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    //CH1: If PUCCHconfigDedicated_v1020 needs to signalled to UE anew then it needs to be configured in SS too

    if (ispresent(v_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.radioResourceConfigDedicated.physicalConfigDedicated.pucch_ConfigDedicated_v1020)) {

      f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_CellConfig_PUCCH_Dedicated(p_TargetPCellId,

                                                                            cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number),

                                                                            v_PcellFDD_TDDInfo.pucch_Config,

                                                                            v_PUCCH_ConfigDedicated));

    }
    // Step 11: Configure target PCell as PCell

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_PCellConfig_REQ(p_TargetPCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), {p_TargetSCellId}, v_CnfFlag));

    // Step 12: Configure target Scell

    if(isvalue (p_SCellToAddModList)) {

      v_SCellIndex := valueof(p_SCellToAddModList[0].sCellIndex_r10);

    } else {

      v_SCellIndex := 1;

    }

    //if Scell change does not happen, steps 1-2 are still needed due to possible change in C-RNTI

    // Save new C-RNTI in cell configuration for p_SCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetSCellId, p_C_RNTI);

    //CH4:   

    if (p_SourcePCellId == p_TargetSCellId and p_SourceSCellId == p_TargetPCellId) 

    {

    f_EUTRA_SS_SRBs_DRBs_Release(p_TargetSCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), f_EUTRA_SS_ActiveDRBs(p_TargetSCellId));

    }
    if ( not v_NoUL_SCell) //@sic R5s140083 sic@

    {

    // Step 12.1: Configure C-RNTI based contention resolution in S-cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetSCellId, p_C_RNTI, v_RachProcedureConfig, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), v_CnfFlag);

    // Step 12.2: configure def UL grant configuration and no auto TA transmision on target Scell

    f_EUTRA_SS_CommonCellConfig(p_TargetSCellId, cas_ULGrantAllocation_REQ(p_TargetSCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), cs_DciInfo_CcchDcchDtchUL_Def, cs_PUCCH_Synch_None, cs_UL_GrantConfig_Def, v_CnfFlag));

    };

    // Step 12.3: Configure SCell as Scell with new Pcell association

    f_EUTRA_SS_CommonCellConfig(p_TargetSCellId, cas_SCellConfig_REQ(p_TargetSCellId, cs_TimingInfo(v_Timing1.SFN.Number, v_Timing1.Subframe.Number), cs_SCellConfig(p_TargetPCellId, v_SCellIndex, p_Scell_Capability, v_DeactTimer,-,-,v_Pcell_Mode), v_CnfFlag));//@sic R5w150205 sic@

    // Step 13:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetPCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));

    // Step 14: configure periodic TA on target cell

    f_EUTRA_SS_CommonCellConfig(p_TargetPCellId, cas_PUCCH_Synch_Config_REQ(p_TargetPCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 15: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_PdcpHandoverComplete_REQ(p_TargetPCellId));

    // Step 16: configure RACH procedure for target cell back to default

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetPCellId, -, -, -, -, -, v_CnfFlag);

    if (p_SourcePCellId != p_TargetSCellId) {  //Source cell is no more a Carrier component

      // Step 17: Re-establish SRBs and release DRB(s) in source cell

      f_EUTRA_SS_SRBs_DRBs_Reset(p_SourcePCellId, -, -, -, v_CnfFlag); // @sic R5s150338: no reconfiguration of DRBs; R5s150922 - Additional changes: f_EUTRA_SS_SRBs_DRBs_Reset sic@

      // Step 18: Release DRX and MeasGap configuration

      f_EUTRA_SS_DrxCtrlConfig(p_SourcePCellId, cs_DrxCtrl_None, -, v_CnfFlag);

      f_EUTRA_SS_MeasGapCtrlConfig(p_SourcePCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);

      //Step 19: Configure source Pcell as normal cell

      f_EUTRA_SS_CommonCellConfig(p_SourcePCellId, cas_ServingCellRelease_REQ(p_SourcePCellId, cs_TimingInfo_Now, v_CnfFlag));

    }

    if (p_SourceSCellId != p_TargetSCellId and p_SourceSCellId != p_TargetPCellId) {

      // No RACH configuration as Already configured for C-RNTI based contention along with Pcell

      //No update of C-RNTI

      //Step 20

      //  Release DRX and MeasGap at the S cell (if necessary)

      if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceSCellId)) {

        f_EUTRA_SS_DrxCtrlConfig(p_SourceSCellId, cs_DrxCtrl_None, -, v_CnfFlag);

      }

      f_EUTRA_SS_MeasGapCtrlConfig(p_SourceSCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);

      //Step 21

      //  Configure SCell as normal cell

      f_EUTRA_SS_CommonCellConfig(p_SourceSCellId, cas_ServingCellRelease_REQ(p_SourceSCellId, cs_TimingInfo_Now, v_CnfFlag));

      //  configure Default grant configuration on S cell

      f_EUTRA_SS_CommonCellConfig(p_SourceSCellId, cas_ULGrantAllocation_REQ(p_SourceSCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, -, v_CnfFlag));

    }

  }
