
	Reason for change:
	Based on 36.331 cl. 5.5.6.1 Actions upon handover and re-establishment the UE shall:

1> release the measurement gaps, if activated;

and based on cl. 5.3.5.4

1> if MAC successfully completes the random access procedure:

     2> apply the parts of the measurement and the radio resource configuration that require the UE to know the SFN of the target PCell (e.g. measurement gaps, periodic CQI reporting, scheduling request configuration, sounding RS configuration), if any, upon acquiring the SFN of the target PCell;
The measurement gap shall be released with the start of HO or reestablishment procedure and shall be activated again after the random access procedure completes. Current TTCN implementation of common HO function may in some test cases leave the gap still configured during the entire HO/Reestablishment procedure and additionally, if the gap was signalled in the HO message, it is being configured to the SS prior transmission of the HO signalling message.

	
	

	Summary of change:
	SS Configuration with the measurement gap has been moved after reception of the signalling message with HO complete (if a gap was configured) and any eventual gap is released in SS as soon as the HO message is sent.

	
	

	Consequences if not approved:

Change 1 
	Function name
	f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_Steps1_12

	Reason for change
	Acc. 36.331 the measurement gap shall be released with the start of HO or reestablishment procedure and shall be activated again after the random access procedure completes. Current TTCN implementation of common HO function may in some test cases leave the gap still configured during the entire HO/Reestablishment procedure and additionally, if the gap was signalled in the HO message, it is being configured to the SS prior transmission of the HO signalling message.


	Summary of change
	SS Configuration with the measurement gap has been moved after reception of the signalling message with HO complete (if a gap was configured) and any eventual gap is released in SS as soon as the HO message is sent.


	TTCN module
	Common\EUTRA\EUTRA_RRCSteps.ttcn

	MCC160 Comment
	


Before change

  function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_Steps1_12(EUTRA_CellId_Type                         p_SourceCellId,

                                                                 EUTRA_CellId_Type                         p_TargetCellId,

                                                                 template (value) DL_DCCH_Message          p_RRC_ConnRecongMsg,

                                                                 template (value) RachProcedureConfig_Type p_RachProcedureConfig,

                                                                 template (value) RadioBearerList_Type     p_DrbConfigList := cs_DrbConfigList_DRB1,

                                                                 C_RNTI                                    p_C_RNTI        := tsc_C_RNTI_Def2,

                                                                 template (omit)  DciUlInfo_Type           p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                                 template (value) UL_GrantConfig_Type      p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                                 template (value) UL_GrantConfig_Type      p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                                 template (present) UL_DCCH_Message        p_RRCConnReconfigComplete := cr_508_RRCConnectionReconfigurationComplete)

    runs on EUTRA_PTC

  { /* @sic R5s120709 function introduced to cope with "Full Handover" as needed for TC_13_4_1_5 sic@ */

    /* @sic R5s140147: additional parameter p_RRCConnReconfigComplete sic@ */

    // @sic R5s151040 sic@

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.measConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    var boolean v_CnfFlag := tsc_NoCnfReq;  /* @sic R5s150922 - Additional changes: v_CnfFlag instead of v_CnfFlag sic@ */

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Step 0: Configuration of DRBs at the target cell

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetCellId, p_DrbConfigList);       /* @sic R5s150338 - dynamic DRB configuration sic@ */

    // Step 1: Transfer of the PDCP Count for AM DRBs from source to target cell   @sic R5-120721 change 7 sic@

    f_EUTRA_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_DRB_IdentityList.AMonly);

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetCellId, p_C_RNTI, p_RachProcedureConfig);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetCellId, p_C_RNTI);

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(p_TargetCellId,

                                                             v_Auth_Params,

                                                             v_NextHopChainingCount,

                                                             -,

                                                             v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

    }
    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    f_IP_ChangeEutraCell(IP, p_TargetCellId);      /* @sic R5s150031 Change 1 sic@ */

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, p_RRC_ConnRecongMsg));

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, p_RRCConnReconfigComplete));       // @sic R5s100430  Change 2, R5s140147 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetCellId, -, -, -, -, -, v_CnfFlag);     /* @sic acc. to R5s100029 change 3 sic@ */

  }
After change

  function f_EUTRA_508RRC_IntraLTE_HO_InterCell_Common_Steps1_12(EUTRA_CellId_Type                         p_SourceCellId,

                                                                 EUTRA_CellId_Type                         p_TargetCellId,

                                                                 template (value) DL_DCCH_Message          p_RRC_ConnRecongMsg,

                                                                 template (value) RachProcedureConfig_Type p_RachProcedureConfig,

                                                                 template (value) RadioBearerList_Type     p_DrbConfigList := cs_DrbConfigList_DRB1,

                                                                 C_RNTI                                    p_C_RNTI        := tsc_C_RNTI_Def2,

                                                                 template (omit)  DciUlInfo_Type           p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                                 template (value) UL_GrantConfig_Type      p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                                 template (value) UL_GrantConfig_Type      p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                                 template (present) UL_DCCH_Message        p_RRCConnReconfigComplete := cr_508_RRCConnectionReconfigurationComplete)

    runs on EUTRA_PTC

  { /* @sic R5s120709 function introduced to cope with "Full Handover" as needed for TC_13_4_1_5 sic@ */

    /* @sic R5s140147: additional parameter p_RRCConnReconfigComplete sic@ */

    // @sic R5s151040 sic@

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var template (omit) MeasConfig v_MeasConfig := p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.measConfig;

    var boolean v_MeasGapConfigIsPresent := isvalue(v_MeasConfig) and ispresent(v_MeasConfig.measGapConfig);

    var DrxCtrl_Type v_DrxCtrl;

    var boolean v_CnfFlag := tsc_NoCnfReq;  /* @sic R5s150922 - Additional changes: v_CnfFlag instead of v_CnfFlag sic@ */

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    // Step 0: Configuration of DRBs at the target cell

    f_EUTRA_SS_CommonRadioBearerConfig(p_TargetCellId, p_DrbConfigList);       /* @sic R5s150338 - dynamic DRB configuration sic@ */

    // Step 1: Transfer of the PDCP Count for AM DRBs from source to target cell   @sic R5-120721 change 7 sic@

    f_EUTRA_SS_PdcpCount_Handover(p_SourceCellId, p_TargetCellId, v_DRB_IdentityList.AMonly);

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure(p_TargetCellId, p_C_RNTI, p_RachProcedureConfig);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI(p_TargetCellId, p_C_RNTI);

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest(p_TargetCellId,

                                                             v_Auth_Params,

                                                             v_NextHopChainingCount,

                                                             -,

                                                             v_DRB_IdentityList.AMonly);

    f_EUTRA_Security_Set(v_Auth_Params);

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    f_IP_ChangeEutraCell(IP, p_TargetCellId);      /* @sic R5s150031 Change 1 sic@ */

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    SRB.send(cas_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, p_RRC_ConnRecongMsg));

    // acc. 36.331, cl. 5.5.6.1 release measurement gap if activated

    f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_None, cs_TimingInfo_Now); 

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, p_RRCConnReconfigComplete));       // @sic R5s100430  Change 2, R5s140147 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure_Def(p_TargetCellId, -, -, -, -, -, v_CnfFlag);     /* @sic acc. to R5s100029 change 3 sic@ */

    if (v_MeasGapConfigIsPresent) {

      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, cs_MeasGapCtrl_Config(valueof(v_MeasConfig.measGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present

    }
  }

